Physical Climatology Problem Set # 2

1. $T_c = 20°C$, $q = 10\ g\,Kg^{-1}$ $p = 1000$ mb, $Z_{top} = 3000$ m

(a) (1)

$$p_{inch} = 1000 \text{ mb} \times 29.92 \text{ inch} / 1013.25 \text{ mb}$$
$$= 29.53 \text{ inch}$$

(2) $$p_{kPa} = 1000 \text{ mb} / (1 \text{ mb} / 0.1 \text{ kPa})$$
$$= 100 \text{ kPa}$$

(3) $$q = 10\ g\,Kg^{-1} = 10 \times 10^{-3}\ Kg\,Kg^{-1} = 0.01\ Kg\,Kg^{-1}$$

(4) $$w = \frac{q}{1-q} = \frac{0.01\ [Kg\,Kg^{-1}]}{1-0.01\ [Kg\,Kg^{-1}]} = 0.010101\ Kg\,Kg^{-1} = 10.10101\ g\,Kg^{-1}$$

(5) $$\frac{w-q}{q} = \frac{10.10101 - 10}{10} = 0.010101 \doteq 1\%$$

The relative difference between w and q is about 1%, meaning in most cases these two variables are interchangeable.

(6) $$T_{°F} = \frac{9}{5} \times T_{°C} + 32 = \frac{9}{5} \times 20 + 32 = 68\ °F$$

(7) $$T_K = T_{°C} + 273.15 = 20 + 273.15 = 293.15\ K$$

(8) $$T_v = T_K(1 + 0.608\,q) = 293.15 \times (1 + 0.608 \times 0.01) = 294.93\ K$$
$$= 21.78\ °C$$

(9) $$R = R_d(1 + 0.608\,q) = 287 \times (1 + 0.608 \times 0.01) \doteq 288.74\ J\,Kg^{-1}K^{-1}$$

(10) In Ideal Gas Law $p = \rho R T$

$$\rho = \frac{p}{RT}$$ where p in Pa, R in $J\,Kg^{-1}K^{-1}$, T in K

$$\therefore \rho = \frac{p_{kPa} \times 1000}{R\,T_K} = \frac{100 \times 1000}{288.74 \times 293.15} \doteq 1.1814\ Kg\,m^{-3}$$

$$\boxed{\text{or } e = \frac{wp}{\varepsilon + w} = \frac{wp}{0.622 + w}}$$

(11) $$\boxed{e = \frac{qp}{0.622 + 0.378q}} = \frac{0.01 \times 1000}{0.622 + 0.378 \times 0.01} = 15.98 \text{ mb}$$

(12) $$\boxed{p_{dry} = p - e} = 1000 - 15.98 = 984.02 \text{ mb}$$

Method 1

(13) $$\boxed{e^* = 6.11 \exp\left[\frac{L}{R_v}\left(\frac{1}{T_f} - \frac{1}{T}\right)\right]} \quad \text{in mb}$$

where $L = 2.501 \times 10^6 \text{ J kg}^{-1}$

$T_f = 273.15$ K

~~T in K~~

$R_v = 461.51 \text{ J kg}^{-1} \text{ K}^{-1}$

$$e^* = 6.11 \times \exp\left[\frac{2.501 \times 10^6}{461.51} \times \left(\frac{1}{273.15} - \frac{1}{293.15}\right)\right]$$

$$= 6.11 \times \exp\left[\frac{2.501 \times 10^6}{461.51} \times \left(3.660992 \times 10^{-3} - 3.411229 \times 10^{-3}\right)\right]$$

$$= 6.11 \times \exp[1.35354 1165] = 6.11 \times 3.87111 \doteq 23.65 \text{ mb}$$

Method 2

$$\boxed{e^* = 6.11 \exp\left[\frac{A(T - T_f)}{T - B}\right]}$$

where $A = \begin{cases} 21.874 & T \le T_f \\ 17.269 & T > T_f \end{cases}$

$B = \begin{cases} 7.66 & T \le T_f \\ 35.86 & T > T_f \end{cases}$

$T_f = 273.15$ K

$\because T > T_f$, $\therefore$ we use $A = 17.269$ and $B = 35.86$

$$e^* = 6.11 \times \exp\left[\frac{17.269 \times (293.15 - 273.15)}{293.15 - 35.86}\right] = 6.11 \times \exp(1.342376)$$

$$= 6.11 \times 3.828130 = 23.3899 \doteq 23.39 \text{ mb}$$

Method 2 is used in subsequent calculations.

(14) $\Delta = e^* - e = 23.39 - 15.98 = 7.41$ mb

(15) $RH = \dfrac{e}{e^*} = \dfrac{15.98}{23.39} = 0.6832 = 68.32\,\%$

(16)
$$q^*_{EXACT} = \frac{0.622\, e^*}{p - 0.378 e^*} = \frac{0.622 \times 23.39}{1000 - 0.378 \times 23.39} = 0.014678 \text{ kg kg}^{-1}$$

$= 14.678$ g kg^{-1} is the result from the exact form.

$$q^*_{APPROX.} = 0.622 \frac{e^*}{p} = 0.622 \times \frac{23.39}{1000} = 0.014548 \text{ kg kg}^{-1}$$

$= 14.548$ g kg^{-1} is the result from the approximate form.

The exact and approx. formulae give different values because the denominators are different in that the approx. form neglects the small term. The simpler form is usually acceptable, but its use can lead to problems when making calculations at different levels in the atmosphere because air pressure falls with height. Look out for this or, for security, always use the exact form.

Note that $q^*_{APPROX.} = q / RH = 10 / 0.6832 = 14.637$ g kg^{-1}
gives the result closer to that from the exact form.

[RH can be approximately defined as $RH \doteq \dfrac{q}{q^*}$, but $RH = \dfrac{e}{e^*}$ is the exact form]

(17) $\because T > T_f$, $A = 17.269$, $B = 35.86$, and $e^* = 6.11 \exp\left[\dfrac{A \times (T - T_f)}{T - B}\right]$

Use $e = 15.98$ to substitute e^* and solve for T, which is the dew point temperature.

$$\therefore T_d = \left(B \ln\frac{e}{6.11} - A T_f\right) / \left(\ln\frac{e}{6.11} - A\right) = \frac{\left(35.86 \times \ln\frac{15.98}{6.11} - 17.269 \times 273.15\right)}{\ln\frac{15.98}{6.11} - 17.269}$$

$$= \frac{-4682.55}{0.961411 - 17.269} = 287.14 \text{ K} = 13.99\ ^\circ\text{C}$$

(18) The latent heat of vaporization of water is a function of temperature, (albeit a weak dependence on temperature).

$$\boxed{L = (2.501 - 0.002361\, T_{°C}) \times 10^6 \quad J\,Kg^{-1}}$$

$$= (2.501 - 0.002361 \times 20) \times 10^6$$

$$= 2453780 \quad J\,Kg^{-1}$$

It is OK to use $L \doteq 2.5 \times 10^6\ J\,Kg^{-1}$ in general applications.

(19) $\because\ q^* = \dfrac{0.622\, e^*}{p - 0.378\, e^*}$

$$\therefore \boxed{\frac{\partial q^*}{\partial T} = \frac{0.622\, p \frac{\partial e^*}{\partial T} - 0.622\, e^* \frac{dp}{dT}}{(p - 0.378\, e^*)^2}}$$

Note: in the class notes or in the book, $\dfrac{\partial q^*}{\partial T} = q^* \dfrac{L}{R_v T^2}$
$\doteq 0.0147 \times \dfrac{2.5\times10^6}{461\times293^2}$
$\doteq 9.26 \times 10^{-4}\ kg\,kg^{-1}\,K^{-1}$

This is an approx. form.

In the above equation, $\boxed{\dfrac{\partial e^*}{\partial T} = \dfrac{A(T_f - B)}{(T-B)^2}\, e^*}$

$$\boxed{\frac{dp}{dT} = \frac{dp}{dz}\frac{dz}{dT} = (-\rho g)\left(-\frac{1}{\Gamma_s}\right) = \frac{gp}{R\Gamma_s T}}$$

This means that Γ_s must be known first. In the class notes, $\boxed{\Gamma_s = \dfrac{g}{C_p + L\frac{\partial q^*}{\partial T}}}$ which itself is dependent upon $\frac{\partial q^*}{\partial T}$. Therefore, we either ignore the second term in the numerator of $\frac{\partial q^*}{\partial T}$, or use a different equation for Γ_s. Let us consider the first option.

Option I: $\dfrac{\partial q^*}{\partial T} \doteq \dfrac{0.622\, p \frac{\partial e^*}{\partial T}}{(p - 0.378\, e^*)^2} = \dfrac{0.622\, p}{(p - 0.378\, e^*)^2} \cdot \dfrac{A(T_f - B)}{(T-B)^2}\, e^*$

$$= \frac{0.622 \times 1000}{(1000 - 0.378 \times 23.39)^2}\ \frac{17.269\,(273.15 - 35.86)}{(293.15 - 35.86)^2} \times 23.39$$

$$\doteq 9.167 \times 10^{-4}\ kg\,kg^{-1}\,K^{-1}$$

option II: This derivation of Γ_s was not discussed in the class. For information, I give the final form of the equation here. According to J.V. Iribarne and W.L. Godson (1981): Atmospheric Thermodynamics (2nd Edition), page 182,

$$\boxed{\Gamma_s = \frac{\Gamma_d \left(\frac{p}{p-e^*}\right)\left[1+\left(\frac{L}{RT}-0.61\right)q^*\right]}{1+\frac{C_{pv}q^*}{C_p}+\frac{L^2 q^*(0.622+q^*)}{C_p R T^2}}}$$

where $C_{pv} = 1850\ \mathrm{J\,kg^{-1}K^{-1}}$, $\boxed{\Gamma_d = \frac{g}{C_p}}$

Substituting all the known values into above equation,

$\Gamma_s \doteq 4.3 \times 10^{-3}\ {}^\circ C/m = 4.3\ {}^\circ C/km$

It should be pointed out that the above equation is the exact form for Γ_s, while $\boxed{\Gamma_s = \frac{g}{C_p + L\frac{\partial q^*}{\partial T}}}$ is the approx. form.

Therefore, $\boxed{\frac{dp}{dT} = \frac{gp}{R\Gamma_s T}} = \frac{9.8 \times 1000}{288.74 \times 4.3 \times 10^{-3} \times 293.15}$

$\approx 26.9253\ \mathrm{mb\ K^{-1}}$

Thus, $\frac{0.622e^*\frac{dp}{dT}}{(p-0.378e^*)^2} = \frac{0.622 \times 23.39 \times 26.9253}{(1000 - 0.378 \times 23.39)^2} = 3.9874 \times 10^{-4}\ \mathrm{Kg\,Kg^{-1}\,K^{-1}}$

The exact form of $\frac{\partial q^*}{\partial T}$ will be

$$\boxed{\frac{\partial q^*}{\partial T} = \frac{0.622p\frac{\partial e^*}{\partial T}}{(p-0.378e^*)^2} - \frac{0.622e^*\frac{dp}{dT}}{(p-0.378e^*)^2}} = (9.167 - 3.987) \times 10^{-4}$$

$= 5.18 \times 10^{-4}\ \mathrm{Kg\,Kg^{-1}\,K^{-1}}$

(20) This question is already addressed, in part, in (19).

$$\Gamma_d = \frac{g}{C_p} = 9.7709168 \times 10^{-3} \ ^\circ C/m \doteq 9.8 \times 10^{-3} \ ^\circ C/m$$

Using the exact form of Γ_s which is introduced in (19) Option II,

$$\Gamma_s \doteq 4.3 \times 10^{-3} \ ^\circ C/m$$

Consider the equation given in the class notes, $\Gamma_s = \frac{g}{C_p + L\frac{\partial q^*}{\partial T}}$.

and use $\frac{\partial q^*}{\partial T}$ from Option I, we get $\Gamma_s \doteq 3.0 \times 10^{-3} \ ^\circ C/m$.

Both the exact and approx. forms can differ by 1.3 °C for every km. The approximate equation gives a smaller saturation lapse rate.

(21) $$\theta = T_K\left(\frac{1000}{P}\right)^{\frac{R}{C_p}} = T_K\left(\frac{1000}{1000}\right)^{\frac{R}{C_p}} = T_K = 293.15\,K = 20\,^\circ C$$

$$\theta_v = T_v\left(\frac{1000}{P}\right)^{\frac{R}{C_p}} = T_v\left(\frac{1000}{1000}\right)^{\frac{R}{C_p}} = T_v = 294.94\,K = 21.79\,^\circ C$$

(b) Let's determine the lifting temperature first. According to the analytical Solution in the class notes,

$$T_{LCL} = \frac{1}{\frac{1}{T_K - 55} - \frac{\ln RH}{2840}} + 55$$

$$= \frac{1}{\frac{1}{293.15 - 55} - \frac{\ln 0.6832}{2840}} + 55$$

$$= \frac{1}{4.199 \times 10^{-3} + 1.3414 \times 10^{-4}} + 55 = 285.78\,K$$

$$= 12.63\,^\circ C$$

Now we can determine the height at the lifting condensation level.

(22) $\because \quad \Gamma_d = \frac{g}{c_p} = -\frac{\partial T}{\partial z} = -\frac{T_{LCL} - T_K}{Z_{LCL} - 0} = \frac{T_K - T_{LCL}}{Z_{LCL}}$

$$\therefore \quad Z_{LCL} = \frac{T_K - T_{LCL}}{\Gamma_d} = \frac{293.15 - 285.78}{9.7709168 \times 10^{-3}} = 754.3 \ m$$

(23) $\because \quad P_{LCL} = P\left(\frac{T_{LCL}}{T_K}\right)^{\frac{g}{R_d \Gamma_d}}$ ⟶ This is a generic relationship between P and T for a constant Γ. For a dry adiabatic process use Γ_d; for a saturated adiabatic process, use Γ_s.

$= P\left(\frac{T_{LCL}}{T_K}\right)^{\frac{C_p}{R_d}}$ ⟶ This is what was given in the class. It is derived from the above with $\Gamma_d = \frac{g}{c_p}$.

$$\therefore \quad P_{LCL} = 1000\left(\frac{285.78}{293.15}\right)^{\frac{1004}{287}}$$

$$= 1000 \times 0.97486^{3.498} = 914.79 \ mb$$

Note that using $R = R_d(1+0.608q)$ to replace R_d, $P_{LCL} = 915.27$ mb.

(24) $\because \quad T_{LCL} = 285.78 \ K = 12.63 \ ^\circ C$

and the air is saturated,

$\therefore \quad T_d = T_{LCL} = 12.63 \ ^\circ C$

$T_v = T_{LCL} \times (1 + 0.608 \times q)$

$= 285.78 \times (1 + 0.608 \times 0.01)$

$= 287.52 \ K = 14.37 \ ^\circ C$

Note that q is the same as that at the base.

(25) During the dry adiabatic uplift from the base to the LCL, θ is conserved, so $\theta_{LCL} = 293.15 \ K = 20 \ ^\circ C$. Also, θ_v is conserved.

$$\theta_{v_{LCL}} = 294.94 \ K = 21.79 \ ^\circ C$$

This can be confirmed by using the equation, (in the case of θ).

$$\theta_{LCL} = T_{LCL}\left(\frac{1000}{P_{LCL}}\right)^{\frac{R_d}{C_p}} = 285.78 \times \left(\frac{1000}{914.79}\right)^{\frac{287}{1004}} = 293.15 K = 20 \ ^\circ C$$

(26) From the base to the LCL, q is conserved.

$$\therefore \; q_{LCL} = 10 \text{ g kg}^{-1}$$

(27)
$$\left.\frac{\partial q^*}{\partial T}\right|_{LCL} = \frac{0.622\, p_{LCL} \left.\frac{\partial e^*}{\partial T}\right|_{LCL}}{(p_{LCL} - 0.378 e^*_{LCL})^2} - \frac{0.622\, e^*_{LCL} \left.\frac{dp}{dT}\right|_{LCL}}{(p_{LCL} - 0.378 e^*_{LCL})^2}$$

$$= 4.12 \times 10^{-4} \text{ Kg Kg}^{-1} \text{ K}^{-1}$$

$$\boxed{\text{or } \frac{\partial q^*}{\partial T} = q^* \frac{L}{R_v T^2} = \frac{0.01 \times 2.5 \times 10^{+6}}{461 \times 285.78^2} \quad \text{Approx.} \approx 6.64 \times 10^{-4} \text{ Kg Kg}^{-1} \text{ K}^{-1}}$$

(28) At the LCL, the air is saturated, so $RH_{LCL} = 100\%$

(29) Using method II in (13),

$$e^*_{LCL} = 6.11 \times \exp\left[\frac{17.269 \times (285.78 - 273.15)}{285.78 - 35.86}\right]$$

$$= 14.62 \text{ mb}$$

Because the air is saturated, the vapor pressure e_{LCL} will be equal to the saturation vapor pressure e^*_{LCL}.

$$\therefore \; e_{LCL} = e^*_{LCL} = 14.62 \text{ mb}.$$

Another approach is to use $e = \dfrac{q\,p}{0.622 + 0.378\, q}$

$$\therefore \; e_{LCL} = \frac{q\, p_{LCL}}{0.622 + 0.378\, q} = \frac{0.01 \times 914.79}{0.622 + 0.378 \times 0.01} = 14.62 \text{ mb}$$

(30) Use the approximate form.
$$\Gamma_{s_{LCL}} = \frac{g}{C_p + L \left.\frac{\partial q^*}{\partial T}\right|_{LCL}} = \frac{9.8}{1004 + 2.5 \times 10^6 \times 4.12 \times 10^{-4}}$$

$$= 4.82 \times 10^{-3} \text{ °C/m}$$

(c) We assume that the saturation adiabatic lapse rate at the LCL remains the same when the air is uplift from the LCL to the top of the moutain.

$$\Gamma_{S_{LCL}} = -\frac{\partial T}{\partial z} = -\frac{T_{Top} - T_{LCL}}{Z_{top} - Z_{LCL}} = \frac{T_{LCL} - T_{Top}}{Z_{Top} - Z_{LCL}}$$

$$\therefore \boxed{T_{Top} = T_{LCL} - \Gamma_{S_{LCL}}(Z_{Top} - Z_{LCL})}$$

$$= 285.78 - 4.82 \times 10^{-3}(3000 - 754.3)$$

$$= 274.96\ K = 1.81\ °C$$

$$\because \boxed{P_{Top} = P_{LCL}\left(\frac{T_{Top}}{T_{LCL}}\right)^{\frac{g}{R_d \Gamma_{S_{LCL}}}}}$$

The derivation was given, for the most part, in the class.
if $\Gamma = -\frac{\partial T}{\partial z}$ = Constant, the P–T relation is determined as follows.

$$T = T_1 - \Gamma z \quad dp = -\rho g\, dz = -\frac{pg}{R_d T} dz$$

$$\frac{dp}{p} = -\frac{g}{R_d}\frac{dz}{T_1 - \Gamma z}, \quad \int_{P_1}^{P} d\ln p = \int_{z_1}^{z} \frac{g}{R_d \Gamma}\frac{d(T_1 - \Gamma z)}{T_1 - \Gamma z}$$

$$\therefore P_{Top} = 914.79 \times \left(\frac{274.96}{285.78}\right)^{\frac{9.8}{287 \times 4.82 \times 10^{-3}}} = 695.94\ mb$$

$$\because e^*_{Top} = 6.11 \times \exp\left[\frac{17.269 \times (274.96 - 273.15)}{274.96 - 35.86}\right]$$

$$= 6.96\ mb$$

and $\boxed{q = \frac{0.622 e}{p - 0.378 e}}$

$$\therefore q_{Top} = \frac{0.622\, e^*_{Top}}{P_{Top} - 0.378\, e^*_{Top}} = \frac{0.622 \times 6.96}{695.94 - 0.378 \times 6.96}$$

$$= 6.244 \times 10^{-3}\ kg\,kg^{-1} = 6.244\ g\,kg^{-1}$$

∴ The amount of liquid water that is condensed during the rise is

$$q_{LCL} - q_{Top} = 10 - 6.244 = 3.756\ g\,kg^{-1}$$ or more properly,

[$W_{LCL} = 10.10101\ g\,kg^{-1}$, $W_{Top} = \frac{q_{Top}}{1 - q_{Top}} = 6.28323\ g\,kg^{-1}$, $\Delta W = 3.81778\ g\,kg^{-1}$]

(d) As shown in the right figure,



$$\Gamma_d = -\frac{\partial T}{\partial z} = -\frac{T_{TOP} - T_{END}}{Z_{TOP} - 0}$$

$$= \frac{T_{END} - T_{TOP}}{Z_{TOP}}$$

$$\therefore \; T_{END} = T_{TOP} + \Gamma_d \cdot Z_{TOP}$$

$$= 274.96 + 9.7709168 \times 10^{-3} \times 3000$$

$$= 304.27 \text{ K}$$
$$= 31.12 \text{ °C}$$

$$P_{END} = P_{TOP}\left(\frac{T_{END}}{T_{TOP}}\right)^{\frac{g}{R_d \Gamma_d}}$$

$$= 695.94 \times \left(\frac{304.27}{274.96}\right)^{\frac{9.8}{287 \times 9.7709168 \times 10^{-3}}}$$

$$= 991.52 \text{ mb}$$

$$e_{END} = \frac{q_{TOP} \times P_{END}}{0.622 + 0.378 \times q_{TOP}} = \frac{6.244 \times 10^{-3} \times 991.52}{0.622 + 0.378 \times 6.244 \times 10^{-3}}$$

$= 9.92$ mb [Note that $q_{END} = q_{TOP} = 6.244 \times 10^{-3}$ kg kg^{-1}]

$$Td_{END} = \frac{35.86 \times \ln\frac{9.92}{6.11} - 17.269 \times 273.15}{\ln\frac{9.92}{6.11} - 17.269} = 280 \text{ K} = 6.85 \text{ °C}$$

$$e^*_{END} = 6.11 \exp\left[\frac{17.269 \times (304.27 - 273.15)}{304.27 - 35.86}\right] = 45.25 \text{ mb}$$

$$RH_{END} = \frac{e_{END}}{e^*_{END}} = \frac{9.92}{45.25} = 0.2192 = 21.92 \%$$

$$q^*_{END} = \frac{0.622\, e^*_{end}}{P_{END} - 0.378\, e^*_{END}} = \frac{0.622 \times 45.25}{991.52 - 0.378 \times 45.25} = 0.02888 \text{ kg kg}^{-1} = 28.88 \text{ g kg}^{-1}$$

$$W^*_{END} = q^*_{END}/(1 - q^*_{END}) = 0.02974 \text{ kg kg}^{-1} = 29.74 \text{ g kg}^{-1}$$

2. Using the pseudoadiabatic chart, see below,



- locate $T=20°C$, $p=1000$ mb (BEGIN), → read $W^* = 15.0\ g\,kg^{-1}$
- locate $W=10.1\ g\,kg^{-1}$, 1000 mb ⟶ read temperature. $T_d \doteq 14\ °C$
- $RH \doteq \frac{W}{W^*} = \frac{10.1}{15.0} \doteq 0.673 = 67.3\%$

(a) ~~Locate the intersect of a dry adiabatic (293.15) and $W = 10.1\ g\,kg^{-1}$~~ which is the LCL, → read $\boxed{Z_{LCL} \doteq 750\ m}$, $\boxed{P_{LCL} \doteq 925\ mb}$ $\boxed{T_{LCL} \doteq 12.5\ °C}$

(b) - Locate the intersect of a saturated adiabatic passing the LCL and $Z = 3000$ m. read $T_{TOP} \doteq 1\ °C$. $P_{TOP} = 700$ mb

(c) - Locate a dry adiabatic passing the top and 1000 mb. read $T_{END} \doteq 31\ °C$, $T_{d\,END} \doteq 6.0\ °C$, $RH_{END} \doteq \frac{6\ g\,kg^{-1}}{30\ g\,kg^{-1}} \doteq 20\%$

(d) Based on the equations described above, a FORTRAN program is written to computer the LCL, the mountain top temperature, and the end temperature and many other variables. The program is accepting an input of T_{base}, P_{base}, q_{base}, Z_{TOP} and dz the increment height upward. Because RH changes little when approaching the LCL, a small increment is recommended for accurately locating the LCL. Also, Γ_s is dependent upon temperature through $\frac{\partial q^*(T,P)}{\partial T}$ which, in turn, depends on height, a small increment is used for accuracy. In a sample printout, see attached $dz = 1$ m is used. The following gives a list of key variables from different methods. The results are very similar.

		Analytical (See equations in this handout)	Pseudoadiabatic Chart	Program (Attached next pages)
Base (Begin)	T_{base}	20°C	20°C	20°C
	P_{base}	1000 mb	1000 mb	1000 mb
	q_{base}	10 g kg^{-1}	10 g kg^{-1}	10 g kg^{-1}
	Tdew	13.99°C	14°C	13.99°C
	θ	20°C	20°C	20°C
LCL	T	12.63°C	12.5°C	12.63°C
	Tdew	12.63°C	12.5°C	12.63°C
	P	914.79 mb	925 mb	915 mb
	Z	754.3 m	750 m	755 m
Top	Z	~~[illegible]~~ 3000 m	3000 m	3000 m
	T	1.81°C	1°C	0.97°C
	Tdew	1.81°C	1°C	0.97°C
	P	695.94 mb	700 mb	697 mb
	θ	31.83°C	31°C	30.90°C
END	T	31.12°C	31°C	30.27°C
	Tdew	6.85°C	6°C	5.98°C
	P	991.52 mb	1000 mb	993 mb
	θ	31.83°C	31°C	30.90°C

Input

ENTER Tbase = 20.0000 in degree Celcius at the base
ENTER Pbase = 1000.000 in mbar at the base
ENTER Qbase = 10.00000 in g/kg at the base
ENTER Ztop = 3000.00 in m for the mountain top height
ENTER dz = 1.00000 in m for the height increment

Output

Z	P	T	Td	q	e	RH	Tv	Thet	Thetv
m	mb	deg C	deg C	g/kg	mb	%	deg C	deg C	deg C
0.	1000.	20.00	13.99	10.00	15.98	68.32	21.78	20.00	21.78
50.	994.	19.51	13.90	10.00	15.89	70.01	21.29	20.00	21.78
100.	988.	19.02	13.81	10.00	15.80	71.76	20.80	20.00	21.78
150.	983.	18.53	13.72	10.00	15.70	73.55	20.31	20.00	21.78
200.	977.	18.05	13.63	10.00	15.61	75.40	19.82	20.00	21.78
250.	971.	17.56	13.54	10.00	15.52	77.30	19.33	20.00	21.78
300.	966.	17.07	13.45	10.00	15.43	79.26	18.83	20.00	21.78
350.	960.	16.58	13.36	10.00	15.34	81.28	18.34	20.00	21.78
400.	954.	16.09	13.27	10.00	15.25	83.36	17.85	20.00	21.78
450.	949.	15.60	13.18	10.00	15.16	85.50	17.36	20.00	21.78
500.	943.	15.12	13.09	10.00	15.07	87.70	16.87	20.00	21.78
550.	938.	14.63	13.00	10.00	14.98	89.98	16.38	20.00	21.78
600.	932.	14.14	12.91	10.00	14.90	92.32	15.89	20.00	21.78
650.	927.	13.65	12.82	10.00	14.81	94.73	15.39	20.00	21.78
700.	921.	13.16	12.73	10.00	14.72	97.22	14.90	20.00	21.78
750.	916.	12.67	12.64	10.00	14.63	99.78	14.41	20.00	21.78
753.	915.	12.65	12.64	10.00	14.63	99.94	14.38	20.00	21.78
754.	915.	12.64	12.63	10.00	14.63	99.99	14.37	20.00	21.78
755.	915.	12.63	12.63	10.00	14.63	100.04	14.36	20.00	21.78
800.	910.	12.41	12.41	9.91	14.41	100.00	14.13	20.23	21.99
900.	900.	11.92	11.92	9.71	13.96	100.00	13.61	20.73	22.47
1000.	889.	11.44	11.44	9.51	13.52	100.00	13.08	21.24	22.94
1100.	878.	10.95	10.95	9.32	13.08	100.00	12.56	21.74	23.41
1200.	868.	10.45	10.45	9.12	12.66	100.00	12.03	22.24	23.88
1300.	858.	9.96	9.96	8.93	12.25	100.00	11.49	22.74	24.35
1400.	847.	9.46	9.46	8.74	11.84	100.00	10.96	23.24	24.81
1500.	837.	8.95	8.95	8.55	11.45	100.00	10.42	23.73	25.28
1600.	827.	8.45	8.45	8.36	11.06	100.00	9.88	24.23	25.74
1700.	817.	7.94	7.94	8.17	10.69	100.00	9.33	24.72	26.20
1800.	807.	7.42	7.42	7.99	10.32	100.00	8.79	25.21	26.66
1900.	798.	6.91	6.91	7.80	9.96	100.00	8.24	25.70	27.12
2000.	788.	6.39	6.39	7.62	9.61	100.00	7.68	26.18	27.57
2100.	778.	5.86	5.86	7.44	9.27	100.00	7.13	26.67	28.02
2200.	769.	5.34	5.34	7.26	8.93	100.00	6.56	27.15	28.47
2300.	760.	4.80	4.80	7.08	8.61	100.00	6.00	27.63	28.92
2400.	750.	4.27	4.27	6.90	8.29	100.00	5.43	28.10	29.37
2500.	741.	3.73	3.73	6.73	7.98	100.00	4.86	28.58	29.81
2600.	732.	3.19	3.19	6.55	7.68	100.00	4.29	29.05	30.25
2700.	723.	2.64	2.64	6.38	7.39	100.00	3.71	29.52	30.69
2800.	714.	2.09	2.09	6.21	7.10	100.00	3.13	29.98	31.12
2900.	706.	1.53	1.53	6.04	6.83	100.00	2.54	30.44	31.56
3000.	697.	0.97	0.97	5.87	6.56	100.00	1.95	30.90	31.98
0.	993.	30.27	5.98	5.87	9.34	21.66	31.36	30.90	31.98
Z	P	T	Td	q	e	RH	Tv	Thet	Thetv
m	mb	deg C	deg C	g/kg	mb	%	deg C	deg C	deg C

Z_LCL = 755.000 m
Rainout = 4.19400 g/kg

```
      PROGRAM RISINGAIR

C_____________________________________________________________________
C
C This program takes input of air temperature, pressure,
C specific humidity and height increment, and then finds
C 1) the lifting condensation level - the height at which a parcel
C    of air becomes saturated,
C 2) the mountain top temperature and pressure, and
C 3) the base temperature, humidity and pressure at the leeside of
C    the mountain
C
C Input:   temperature, pressure, specific humidity, height of
C          mountain top, height increment
C
C Physics: the rising air first follows the dry adiabatic lapse
C          rate until the LCL is reached, then follows the
C          saturated adiabatic lapse rate to the top of the
C          mountain is reached, and then follows the dry adabatic
C          lapse rate to the base level.
C
C Output: P,T,Td,q,e,RH,Tv,Thet,Thetv
C
C Author: Zong-Liang Yang 3/31/2003
C ____________________________________________________________________
C
      IMPLICIT NONE
C Input variables
C
      REAL TBASE, PBASE, QBASE, ZTOP, DZ
C
C PHYSICAL CONSTANTS
C
      REAL CP, RD, G, EPS, CPORD, GD, L
C
C INTERMEDIATE VARIABLES
C
      REAL RA, GS, TOLD, DQSDT, DPDT, DESDT, WOLD
C
C OUTPUT VARIABLES
C
      REAL T, T0, Z, P, Q, W, E, ES, RH, ZLCL
      REAL TD,TV,THET,THETV
      REAL RAINOUT
C
C ASSIGN THE VALUES TO THE CONSTANTS
C
      CP = 1004.0 !Specific Heat of dry air, in J/K/kg
      Rd = 287.0  !Gas constant for dry air, in J/K/kg
      EPS = 0.622 !ratio of the mass of water vapor to the mass of dry air
                  ! (Mv/Md)
      G   = 9.80665 !gravity, in m/s**2
      L   = 2.501e06 !latent heat of vaporization, J/kg

C
C Save output variables in a file
C
      OPEN(11, file ='LCL_MTOP.dat', status='unknown')
C
C READ INPUT
C
      WRITE(6,*) 'ENTER Tbase = ', ' in degree Celcius at the base'
      READ(5,*) TBASE
      WRITE(11,*) 'ENTER Tbase = ',TBASE,' in degree Celcius
     &             at the base'

      WRITE(6,*) 'ENTER Pbase = ', ' in mbar at the base'
      READ(5,*) PBASE
      WRITE(11,*) 'ENTER Pbase = ',PBASE,' in mbar at the base'

      WRITE(6,*) 'ENTER Qbase = ', ' in g/kg at the base'
      READ(5,*) QBASE
```

```
      WRITE(11,*) 'ENTER Qbase = ',QBASE, ' in g/kg at the base'

      WRITE(6,*) 'ENTER Ztop  = ', ' in m for the mountain top height'
      READ(5,*) ZTOP
      WRITE(11,*) 'ENTER Ztop  = ',ZTOP, ' in m for the mountain top
     &             height'

      WRITE(6,*) 'ENTER dz  = ', ' in m for the height increment'
      READ(5,*) DZ
      WRITE(11,*) 'ENTER dz  = ',DZ, ' in m for the height increment'

C _____________________________________________________________
C Step 1: determine the LCL
C _____________________________________________________________

C Compute upwards from base of mountain
      Z = 0                    !kilometers above base level
      T0 = TBASE+273.15        !degree C to Kelvin
      GD = G/CP                !dry adiabatic lapse rate
      CPORD = CP/RD            !for Poisson equation
      Q  = QBASE/1000.0        !g/kg to kg/kg
C
C
C
      WRITE(11,2000)
      WRITE(11,2001)
1     CONTINUE
         T = T0-Z*GD
         RA    = RD*(1.0+0.608*Q)
         P = PBASE*(T/T0)**(G/RA/GD)
         E = Q*P/(EPS+(1.-EPS)*Q)
         CALL ESAT(T,ES,DESDT) !Subroutine to calculate esat, desat/dT
         CALL TDEW(T,E,TD)     !Subroutine to calculate TD
         RH = (E/ES)*100.0
         TV = (1.+0.608*Q)*T
         THET = T*(1000.0/P)**(RA/CP)
         THETV = TV*(1000./P)**(RA/CP)
C
C OUTPUT EVERY 50 meters or |RH-100| < 0.1
C
         IF (MOD(Z,50).EQ.0 .OR. ABS(RH-100).LT.0.1)
     &     WRITE(11,2002) Z,P,T-273.15,TD-273.15,Q*1000.,E,RH,TV-273.15,
     &     THET-273.15,THETV-273.15
C
C Now move up one height increment
C
         Z = Z + DZ

         IF (RH.LE.100.0) THEN
            Q = QBASE/1000.0
            GOTO 1
         ELSE
            Z = Z - DZ
            GOTO 99
         ENDIF

2000   FORMAT(6x,'Z',6x, 'P',6x,'T',5x,'Td',6x,'q',6x,'e',5x,'RH',
     &          5x,'Tv',3x,'Thet',2x,'Thetv')
2001   FORMAT(6x,'m',5x,'mb',2x,'deg C',2x,'deg C',3x,'g/kg',5x,'mb',
     &  6x,'%',2x,'deg C',2x,'deg C',2x,'deg C')
2002   FORMAT(2F7.0,9F7.2)

C
C LCL FOUND
C
99       CONTINUE

         ZLCL = Z
C
C Now find the mountain top pressure, temperature, humidity, ...
C
```

```
        WOLD = Q/(1.0-Q)  ! mixing ratio in kg/kg
        GS   = GD         ! initialize GS for the first increment of height up

1111    CONTINUE

         RA    = RD*(1.0+0.608*Q)
         DPDT  = G*P/(RA*T*GS)
         CALL ESAT(T,ES,DESDT)       !Subroutine to calculate esat,dest/dT
         DQSDT = (DESDT*EPS*P-EPS*ES*DPDT)/(P-0.378*ES)**2.0
         GS    = G/(CP+L*DQSDT)
C
C Now add an increment of height, DZ
C
         Z    = Z + DZ
         TOLD = T
C
C Now find T, P for the new height with the increment dz
C
         T = T - DZ*GS
         P = P*(T/TOLD)**(G/RA/GS)  !Class note P=f(T) given constant a GS

C
C As the air is saturated now, and es is known from T, we can find q from
C
         CALL ESAT(T,ES,DESDT)       !Subroutine to calculate esat,dest/dT
         Q = EPS*ES/(P-(1.0-EPS)*ES)
         TD = T

         RH = 100.0
         E  = ES

         TV = (1.+0.608*Q)*T
         THET = T*(1000.0/P)**(RA/CP)
         THETV = TV*(1000./P)**(RA/CP)
C
C OUTPUT EVERY 100 meters
C
         IF (MOD(Z,100).EQ.0)
     &    WRITE(11,2002) Z,P,T-273.15,TD-273.15,Q*1000.,E,RH,TV-273.15,
     &    THET-273.15,THETV-273.15

C
C Repeat calculation for next height increment
C
       IF (Z.LT.ZTOP) GOTO 1111

999    CONTINUE

C
C find out how much water has been rained out
C
       W = EPS*ES/(P-ES)
       RAINOUT = WOLD - W

C
C If the air is brought back to the base dry adiabatically, then T, P,
C
         Z    = Z - ZTOP
         TOLD = T
         T = T+GD*ZTOP

         RA    = RD*(1.0+0.608*Q)
         P = P*(T/TOLD)**(G/RA/GD)  !
         E = Q*P/(EPS+(1.-EPS)*Q)
         CALL ESAT(T,ES,DESDT) !Subroutine to calculate esat, desat/dT
         CALL TDEW(T,E,TD)     !Subroutine to calculate TD
         RH = (E/ES)*100.0
         TV = (1.+0.608*Q)*T
         THET = T*(1000.0/P)**(RA/CP)
         THETV = TV*(1000./P)**(RA/CP)
C
C OUTPUT
```

```
C
          WRITE(11,2002) Z,P,T-273.15,TD-273.15,Q*1000.,E,RH,TV-273.15,
     &             THET-273.15,THETV-273.15
          WRITE(11,2000)
          WRITE(11,2001)
C
C
        WRITE(11,*) 'Z_LCL   = ', ZLCL, '  m'
        WRITE(11,*) 'Rainout = ', RAINOUT*1000.0, '  g/kg'
C
C End of main program
C
        STOP
        END

        SUBROUTINE ESAT(T,ES,DESDT)
C
C Determine ES, given T
C
        IMPLICIT NONE
        REAL T,ES,A,B,TF,DESDT
        TF = 273.15
        IF (T .LE. TF) THEN
                 A = 21.874
                 B = 7.66
        ELSE
                 A = 17.269
                 B = 35.86
        END IF

        ES    = 6.11*EXP((A*(T-TF))/(T-B))
        DESDT = A*(TF-B)*ES/(T-B)**2.0
        RETURN
        END

        SUBROUTINE TDEW(T,E,TD)
C
C Determine TD, given T and E
C
        IMPLICIT NONE
        REAL T,E,A,B,TF,TD
        TF = 273.15
        IF (T .LE. TF) THEN
                 A = 21.874
                 B = 7.66
        ELSE
                 A = 17.269
                 B = 35.86
        END IF

        TD = (B*log(E/6.11)-A*TF)/(log(E/6.11)-A)
        RETURN
        END
```